

钢筋截料优化

张长源

(水电部第十工程局)

一、问题的提出

大、中型水电站使用的钢筋量少则千吨,多则万吨,如果能减少钢筋耗费,降低水电站成本是显著的。减少钢筋耗费,一是加强管理,杜绝浪费;二是从技术上采用较先进的计算方法,充分利用钢筋强度。减少钢筋截料时的弃料,是节约钢筋的主要途径之一。

在钢筋制作过程中,根据工程设计的钢筋图和钢筋进料情况,在满足于规范要求的钢筋搭头位置、下料长度等的前提下,尽量减少钢筋下料后的弃料数量。当一种规格的钢筋需要截成不同长度的几种料时,往往有多种不同的截料组合方案,而截料组合方案的数量与钢筋的截料种类和截料长度有关,截料种类越多,截料的组合方案越多,截料长度与钢筋进料长度之差越大时,截料的组合方案也越多,带有很大的随机性。在一般配料中都采用试算法,繁琐费时。笔者曾试算过两个实例(表1)。

表 1

实例 编号	进料长度 (厘米)	截料种类 长度(厘米)							截料组合 方案数
		1	2	3	4	5	6	7	
1	200	150	100	50	25	10			39
2	400	230	210	140	110	80	50	30	103

可见截料组合方案多,不仅人力难以胜任,且最终结果不一定最优。因此,可以用线性优化方法得出最优结果。关于线性规划的解算方法和计算机程序已见于不少著作中,可直接运用。但前已述及,钢

筋截料时,需先试算出有多少种可供选择的截料组合方案(这部分工作一般需要人工完成),将各截料组合方案的弃料值作为目标函数输入优化程序中进行解算,以得出最优结果。当截料组合方案数太大时,而需人工试算部份往往容易出错,费时亦多,施工单位也很少采用。鉴于此,笔者用 BASIC 语言编写钢筋截料组合方案的试算法程序,并使之与线性优化程序相结合。当按要求输入有关的参数后,计算机可以直接算出钢筋截料的所有组合方案,优化后的各方案变量取值,以及六个参数:总使用长度、总使用根数、实际使用重量、包括损耗在内的使用总重量、弃料损耗率、包括钢筋设计搭头在内的损耗率。

二、钢筋截料组合方案的试算法

钢筋截料的组合一般指一种规格的钢筋需截取两种或两种以上的长度而言,若是截一种长度的料,即只有一种截法。钢筋截料的组合只能采用试算法,但试算却有一定的

规律可循，可编排成简单的计算机程序进行计算。例如，有一批长度为180厘米的钢筋，需要截成三种长度：72厘米、52厘米、35厘米，求有多少种截法。根据试算有八种截法（表2）：

表 2

根数 长度(厘米)	(一)	(二)	(三)	(四)	(五)	(六)	(七)	(八)
70	2	1	1	1	0	0	0	0
52	0	2	1	0	3	2	1	0
35	1	0	1	3	0	2	3	5
余料	5	6	23	5	24	6	23	5

当试算有多少种截料组合时，先把所截不同长度的料从大到小排列，再从最长的一种料开始截取。

从截法（一）可以看出，70厘米长的料最多只能截取2根，也就是说在不同的截法中，70厘米长度的料的截取根数可以在0~2的范围内变化。同理，当截取52厘米长度时，最多可截取3根，截取根数在0~3的范围内变化，对于最后一种料的截取根数，取决于前面两种料组合截取后的余料长度。

可以看出，进行截料组合的试算法，有四个要点：

- （1）不同长度料的截取，应从大到小的顺序排列。
- （2）一种料的截取根数，取决于前面截取后的余料。
- （3）每一种料截取根数在零至截取的最大根数之间变化。

（4）截料组合方案（一）采用从大到小的顺序计算，以后的各方案采用递推的方法循环计算。

本文提供的计算程序即按照以上几点编写而成，在工程实际运用中，一种规格的钢筋同时截料的种类一般以3~5种居多，故本文提供的计算程序截料种类限制在2~10种，可满足电站施工中单项工程的钢筋截料问题。

三、钢筋截料优化的程序框图及程序

（一）钢筋截料优化的程序框图见图1；

（二）钢筋截料优化的BASIC语言程序。

使用本程序时，应输入下列数据：

R——截料种类；

S(I)——钢筋进料长度（厘米），I是与R有关的序号；

G——钢筋每米重量（公斤/米）。

DM——钢筋设计搭接头数量（个）；

DL——钢筋设计搭接头长度（厘米）；

SU——钢筋设计总使用长度（米）；




图 1 程序框图

NL——按钢筋进料长度使用的根数；

F(I)——钢筋截料长度（厘米），I 是与 R 有关的序号；

B(I)——与 F(I) 相应的需要数量（根）。

本程序的输出结果有以下五种：

1. 当钢筋进料长度小于所需截料长度中的最大值时，即打印出“LENGTH, ERROR”；

2. 当钢筋长度和排列无差错时，打印出截料组合方案“SCHEME”以及相对应的边料长度，“SURPLUS”；

- 3 当优化问题无解时, 打印出 "INFEASIBLE" ;
4. 当优化问题无约束时, 打印出 "UNBOUNDED" ;
5. 当优化问题有解时即打印出

X(I) —— 结构变量的最优解值(根),

SUM. SURPLUS. —— 目标函数最优解值(余料总量, 厘米),

SUM. USE. LENGTH. —— 钢筋使用总长度(米),

SUM. NUMBER. —— 钢筋总根数(由两个数组成, 前一个数为按钢筋进料长度使用的根数; 后一个数为结构变量最优解值的累加数),

USE. WEIGHT. —— 钢筋实用重量(吨),

SUM. WEIGHT. —— 包括损耗在内的钢筋总重量(吨),

SURPLUS. LOSS. RATE —— 余料损耗率,

SUM. LOSS. RATE. —— 包括钢筋搭接头在内的损耗率。

具体程序如下:

```

10 PRINT "INPUT R,S(I),G,DM,DL,S,U,NL"
12 PRINT "R=?"
14 INPUT R
16 DIM S(12)
18 PRINT "S(11-R)=?"
20 INPUT S(11-R)
22 T(1)=S(11-R)
24 PRINT "G=?"
26 INPUT G
28 PRINT "DM=?"
30 INPUT DM
32 PRINT "DL=?"
34 INPUT DL
36 PRINT "SU=?"
38 INPUT SU
40 PRINT "NL=?"
42 INPUT NL
44 PRINT
46 PRINT "INPUT F(I),B(I)"
48 FOR I=11-R TO 10
50 PRINT "F(I)=? B(I)=?"
52 INPUT F(I),B(I)
54 NEXT I
60 FOR I=11-R TO 9
65 FOR J=1+1 TO 10

```

```
70 IF F(I) >= F(J) THEN 90
75 A1=F(I):A2=B(I)
80 F(I)=F(J):B(I)=B(J)
85 F(J)=A1:B(J)=A2
90 NEXT J
95 NEXT I
100 PRINT
105 T(2)=F(11-R)
110 DIM H(11,20*R)
115 IF T(1) < T(2) THEN 125
120 GOTO 135
125 PRINT "LENGTH,ERROR"
130 GOTO 1360
135 T=1
140 ON R-1 GOTO 305, 285, 265, 245, 225, 205, 185, 165
145 Y(1)=INT (S(1)/F(1))
150 FOR A=0 TO Y(1)
155 C(1)=Y(1)-A
160 S(2)=S(1)-F(1)*C(1)
165 Y(2)=INT (S(2)/F(2))
170 FOR B=0 TO Y(2)
175 C(2)=Y(2)-B
180 S(3)=S(2)-F(2)*C(2)
185 Y(3)=INT (S(3)/F(3))
190 FOR C=0 TO Y(3)
195 C(3)=Y(3)-C
200 S(4)=S(3)-F(3)*C(3)
205 Y(4)=INT (S(4)/F(4))
210 FOR D=0 TO Y(4)
215 C(4)=Y(4)-D
220 S(5)=S(4)-F(4)*C(4)
225 Y(5)=INT (S(5)/F(5))
230 FOR E=0 TO Y(5)
235 C(5)=Y(5)-E
240 S(6)=S(5)-F(5)*C(5)
245 Y(6)=INT (S(6)/F(6))
250 FOR F=0 TO Y(6)
255 C(6)=Y(6)-F
```

```
260 S(7)=S(6)-F(6)*C(6)
265 Y(7)=INT ( S(7)/F(7) )
270 FOR G=0 TO Y(7)
275 C(7)=Y(7)-G
280 S(8)=S(7)-F(7)*C(7)
285 Y(8)=INT ( S(8)/F(8) )
290 FOR H=0 TO Y(8)
295 C(8)=Y(8)-H
300 S(9)=S(8)-F(8)*C(8)
305 Y(9)=INT ( S(9)/F(9) )
310 FOR I=0 TO Y(9)
315 C(9)=Y(9)-I
320 S(10)=S(9)-F(9)*C(9)
325 C(10)=INT ( S(10)/F(10) )
330 S(11)=S(10)-F(10)*C(10)

335 PRINT "SCHEME("T")" ;
340 FOR J=11-R TO 10
345 H(J, T)=C(J)
350 PRINT H(J, T)* " ";
355 NEXT J
360 H(11, T)=S(11)
365 PRINT "SURPLUS. =" H(11, T)
370 T=T+1
375 NEXT I
380 IF R=2 THEN 460
385 NEXT H
390 IF R=3 THEN 460
395 NEXT G
400 IF R=4 THEN 460
405 NEXT F
410 IF R=5 THEN 460
415 NEXT E
420 IF R=6 THEN 460
425 NEXT D
430 IF R=7 THEN 460
435 NEXT C
440 IF R=8 THEN 460
```

```
445 NEXT B
450 IF R=9 THEN 460
455 NEXT A
460 M=T-1
465 N=R
470 P=-1
475 DIM A(N+2,N+M+2),X(M),L(N+1)
480 FOR J=1 TO M
485 A(1,J)=H(11,J)
490 NEXT J
495 FOR I=2 TO N+1
500 A(I,M+1)=0
505 NEXT I
510 FOR I=2 TO N+1
515 A(I,M+2)=B(I+9-N)
520 NEXT I
525 ON N-1 GOTO 805,775,745,715,685,655,625,595
565 FOR I=2 TO N+1
570 FOR J=1 TO M
575 A(I,J)=H(I-1,J)
580 NEXT J
585 NEXT I
590 GOTO 830
595 FOR I=2 TO N+1
600 FOR J=1 TO M
605 A(I,J)=H(I,J)
610 NEXT J
615 NEXT I
620 GOTO 830
625 FOR J=2 TO N+1
630 FOR J=1 TO M
635 A(I,J)=H(I+1,J)
640 NEXT J
645 NEXT I
650 GOTO 830
655 FOR I=2 TO N+1
660 FOR J=1 TO M
665 A(I,J)=H(I+2,J)
```

```
670 NEXT J
675 NEXT I
680 GOTO 830
685 FOR I=2 TO N+1
690 J=1 TO M
695 A(I,J)=H(I+3,J)
700 NEXT J
705 NEXT I
710 GOTO 830
715 FOR I=2 TO N+1
720 FOR J=1 TO M
725 A(I,J)=H(I+4,J)
730 NEXT J
735 NEXT I
740 GOTO 830
745 FOR I=2 TO N+1
750 FOR J=1 TO M
755 A(I,J)=H(I+5,J)
760 NEXT J
765 NEXT I
770 GOTO 830
775 FOR I=2 TO N+1
780 FOR J=1 TO M
785 A(I,J)=H(I+6,J)
790 NEXT J
795 NEXT I
800 GOTO 830
805 FOR I=2 TO N+1
810 FOR J=1 TO M
815 A(I,J)=H(I+7,J)
820 NEXT J
825 NEXT I
830 FOR I=2 TO N+1
835 FOR J=1 TO M+2
840 A(I,J)=A(I,J)*SGN ( A(I,M+2) )
845 NEXT J
850 NEXT I
855 E=0 0001
```



```
860 FOR I=1 TO M
865 A(1,I)=-P•A(1,I)
870 NEXT I
875 IF N=1 GOTO 915
880 FOR I=2 TO N+1
885 A(I,N+M+1)=A(I,M+2)
890 A(I,M+2)=0
895 IF I=2 GOTO 910
900 A(I,M+I-1)=A(I,M+1)
905 A(I,M+1)=0
910 NEXT I
915 R=1
920 FOR I=2 TO N+1
925 L(I)=M+I-1
930 IF A(I,M+I-1)=1 GOTO 960
935 L(I)=N+M+1
940 FOR J=1 TO N+M
945 A(N+2,J)=A(N+2,J)-A(I,J)
950 NEXT J
955 R=N+2
960 NEXT I
965 T=1
970 FOR I=2 TO N+M
975 IF A(R,I)-A(R,T)>E GOTO 1000
980 IF A(R,I)-A(R,T)<-E GOTO 995
985 IF R=1 GOTO 1000
990 IF A(1,I)-A(1,T)>=-E GOTO 1000
995 T=I
1000 NEXT I
1005 IF A(R,T)<-E GOTO 1065
1010 IF R=1 GOTO 1195
1015 FOR I=1 TO N+M
1020 IF A(R,I)>E GOTO 1040
1025 NEXT I
1030 R=1
1035 GOTO 965
1040 FOR I=2 TO N+1
1045 IF L(I)<=N+M GOTO 1055
```

```
1050 IF A(I,N+M+1)>E GOTO 1175
1055 NEXT I
1060 GOTO 1195
1065 S=1
1070 FOR I=2 TO N+1
1075 IF A(I,T)<=E GOTO 1100
1080 Y=A(I,N+M+1)/A(I,T)
1085 IF S=1 GOTO 1095
1090 IF Y>=A(S,N+M+1)/A(S,T) GOTO 1100
1095 S=I
1100 NEXT I
1105 IF S=1 GOTO 1185
1110 L(S)=T
1115 Y=A(S,T)
1120 FOR I=1 TO N+M+1
1125 A(S,I)=A(S,I)/Y
1130 NEXT I
1135 FOR I=1 TO N+2
1140 IF I=S GOTO 1165
1145 Y=A(I,T)
1150 FOR J=1 TO N+M+1
1155 A(I,J)=A(I,J)-Y*A(S,J)

1160 NEXT J
1165 NEXT I
1170 GOTO 965
1175 PRINT "INFEASIBLE"
1180 GOTO 1360
1185 PRINT "UNBOUNDED"
1190 GOTO 1360
1195 FOR I=2 TO N+1
1200 IF L(I)>M GOTO 1215
1205 S=L(I)
1210 X(S)=A(I,N+M+1)
1215 NEXT I
1220 Y=P*A(1,N+M+1)
1225 PRINT
1230 FOR I=1 TO M
```

```
1235 PRINT "X(", I; ")="; X(I); " ROOT"
1240 NEXT I
1245 PRINT
1250 PRINT "SUM,SURPLUS,="; Y; "CM"
1255 PRINT
1260 PRINT "SUM,USE,LENGTH,="; SU; "M"
1265 PRINT
1270 A=DM*DL
1275 B=0
1280 FOR I=1 TO M
1285 B=B+X(I)
1290 NEXT I
1295 PRINT "SUM,NUMBER,="; DM; " +"; B" ROOT"
1300 PRINT
1305 D=(SU*G)/1000
1310 PRINT "USE,WEIGHT,="; D; " TON"
1315 PRINT
1320 C=((Y+A)/100+SU)*G/1000
1325 PRINT "SUM,WEIGHT,="; C; " TON"
1330 PRINT
1335 E=Y/(Y+SU*100+A)
1340 PRINT "SURPLUS,LOSS,RATE,="; E
1345 PRINT
1350 F=(Y+A)/(Y+SU*100+A)
1355 PRINT "SUM,LOSS,RATE,="; F
1360 END
```

四、算 例

以一水电站首部枢纽防渗混凝土铺盖为例，其钢筋表和材料表与工地实际使用的钢筋长度见表3、4、5。

表3所示使用 $\phi 30$ 钢筋的有两种长度，使用 $\phi 22$ 钢筋的有4种长度，使用 $\phi 16$ 钢筋的有5种长度。但大多数型式的钢筋长度都大于钢筋实有长度，这就需要数根钢筋焊接而成，接头所需长度因不同焊接方式而异，本处采用单面搭接焊，接头长度按八倍钢筋直径计算。下面仅以 $\phi 30$ 钢筋的计算情况来说明整个计算的过程。

$\phi 30$ 钢筋组合的计算数据整理如下：

1. 第一种钢筋配料数据

下料长度：2016厘米。

所需根数: 23 根

钢筋实长: 850 厘米

接头长度: $3 \times 8 = 24$ 厘米,

配料方式: $2 \times (850 - 24) + 364 = 2016$ 厘米

按实长使用的钢筋根数: $2 \times 23 = 46$ 根

需截取的钢筋长度及根数: 364 厘米, 23 根。

接头个数: $2 \times 23 = 46$ 个

2. 第二种钢筋配抖数据

表 3

编号	直径	型 式	长 度 (米)	根 数	总 长 (米)
1	Ø30		20.16	23	463.68
2	Ø30		17.16	37	634.92
3	Ø16		11.98	154	1844.92
4	Ø16		4.58	50	229.00
5	Ø22		11.58	23	266.34
6	Ø22		14.03	60	841.80
7	Ø22		8.58	37	317.46
8	Ø22		9.81	60	588.60
9	Ø16		7.40	33	244.20
0	Ø16		5.18	30	155.40
	Ø16		8.00	16	128.00

下料长度: 1716 厘米

所需根数: 37 根

钢筋实长: 850 厘米

接头长度: $3 \times 8 = 24$ 厘米

配料方式: $2 \times (850 - 24) + 64 = 1716$ 厘米

按实长使用的钢筋根数: $2 \times 37 = 74$ 根

需截取的钢筋长度及根数: 64 厘米长 37 根

接头个数: $2 \times 37 = 74$ 个

表 4

直径	总 长 (米)	单位重 (公斤/米)	总 重 (公斤)	计入5% 总重 (公斤)	混凝土 (米³)
Ø16	2601.52	1.578	4105.20	4310.50	779
Ø22	2014.20	2.984	6010.37	6310.37	
Ø30	1098.60	5.549	6096.13	6400.94	
合计	含钢量20.81公斤/米³		16211.70	17022.33	779

表 5

编 号	规 格	长 度 (米)
1	Ø30	8.5
2	Ø22	7.0
3	Ø16	7.5

3. 计算输入数据

 $R=2, S(I)=850, G=5.549, DM=120, DL=24,$
 $SU=1098.6, NL=120$

F(I) B(I)

364 23

64 37

计算结果如下:

IRUN

INPUT R,S(I),G,DM,DL,SU,NL

R=?

?2

S(9)=?

?850

G=?

?5.549

DM=?

?120

DL=?

?24

SU=?

?1098.6

NL=?

?20

INPUT F(I),B(I)

F(9)=? B(9)=?

?64

??23

F(10)=? B(10)=?

?64

??37

SCHEME(1) 2 1 SURPLUS.=58

SCHEME(2) 1 7 SURPLUS.=38

SCHEME(3) 0 13 SURPLUS.=18

X(1)=11.5 ROOT

X(2)=0 ROOT

X(3)=1.96153846 ROOT

```

SUM_SURPLUS.=702.307693 CM
SUM_USE_LENGTH.=1098.6 M
SUM_NUMBER.=120+13.4615385 ROOT
USE_WEIGHT.=6.0961314 TON
SUM_WEIGHT.=6.29491365 TON
SURPLUS_LOSS_RATIO.=6.19087981E-03
SUM_LOSS_RATE.=.0315782336

```

五、小 结

通过以上计算实例可以看出,需要截料的钢筋长度数据,与人为的配料有关,当所截钢筋中长度最小的一种料长度较大时,截下的余料仍可进行新的配料,而不做弃料处理。例如表3中的 $\phi 16$ 钢筋就有类似的情况,即截下的两根余料可以焊成一根长度合适的料。对于钢筋的配料,可以根据各种不同的情况,进行不同的配料组合,以达到节约钢筋的目的。

本文程序适用的截料种类范围为2~10种。因为截料种类为1时,只有一种结果;截料种类大于10,在水电施工中尚不多见,为使程序简短,只考虑能满足工地常用的一般情况。程序中的钢筋使用数量,在约束条件中均采用了“=”的运算符,这样有的题目由于约束条件没有共同的交点,会出现无解的情况,这时,有两点可参考:

1. 钢筋截料组合方案部份可以作为截料的参考数据。

2. 将程序中的500语句由“ $500A(I, M+1)=0$ ”改为“ $500A(I, M+1)=-1$ ”。这时钢筋用量的约束条件变为“ $\geq$ ”的运算符。如果计算出的变量解值与实际要求数量相差不大,其变量解值则认为是可行的最优解。如果计算出的变量解值与实际要求数量相差较大,其变量解值仅能作为参考,这时,其它参数的计算结果不再符合实况。

关于使用钢筋的损耗,主要由两部份组成。一是钢筋截料时的边角料;二是钢筋搭接头消耗,其消耗数量与钢筋接头型式和焊接工艺有关。过去钢筋的设计损耗率按4%计。近年来由于某种原因,钢筋的设计损耗率按5%计,是导致钢筋损耗增加的原因之一。笔者采用本文的程序用APPLE II微机进行了计算。在有解的情况下,钢筋总损耗率在3%左右,其中大部份是钢筋接头消耗,弃料很少。如果我们能对钢筋截料推行优化计算,能减少1%的钢筋损耗,对于一个水电站的建造来说,已受益匪浅了。